# Maintainer: Alexey Pavlov <alexpux@gmail.com>

_realname=XML-SAX-Base
pkgname=perl-${_realname}
pkgver=1.09
pkgrel=1
pkgdesc="Base class SAX Drivers and Filters"
arch=('any')
url="https://metacpan.org/release/XML-SAX-Base"
groups=('perl-modules')
license=('GPL' 'PerlArtistic')
depends=('perl')
options=('!emptydirs')
source=("https://www.cpan.org/authors/id/G/GR/GRANTM/${_realname}-${pkgver}.tar.gz")
sha256sums=('66cb355ba4ef47c10ca738bd35999723644386ac853abbeb5132841f5e8a2ad0')

build() {
  cd ${srcdir}/${_realname}-${pkgver}
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd  ${srcdir}/${_realname}-${pkgver}
  make test
}

package() {
  cd  ${srcdir}/${_realname}-${pkgver}
  make install DESTDIR=$pkgdir

  find ${pkgdir} -name '.packlist' -delete
  find ${pkgdir} -name '*.pod' -delete
}
